

# Using Functions SYSFUNC and IFC to Conditionally Execute Statements in Open Code

Ronald J. Fehd, Centers for Disease Control and Prevention, Atlanta, GA, USA

## ABSTRACT

**Description :** This paper explains how to conditionally execute statements in open code without having to wrap the code in a macro. This is accomplished by combining the macro function `sysfunc` with the data step function `ifc`. The result is that parameterized include programs gain the power of conditional processing — `%if ... %then` — for which macros are usually used.

**Purpose :** replacement for macro if statement

**Audience :** advanced users and macro programmers.

**Keywords :** assertions, function, ifc, nrstr, parameterized include program, sysfunc

**Information :** assertions, conditional processing, parameterized include programs

## Contents

<b>Functions</b>	<b>2</b>
Data Step Function IFC . . . . .	3
Macro Function SysFunc . . . . .	3
Combining SysFunc and IFC . . . . .	4
<b>Usage Examples</b>	<b>5</b>
Calling Routines . . . . .	6
Assertions . . . . .	9
<b>Conclusion</b>	<b>10</b>

---

## Introduction

### Overview

Program development proceeds through these stages:

- create ad hoc programs
- identify programs with similar algorithms
- consolidate into parameterized include program
- where necessary, convert to macro for conditional processing

The techniques shown here eliminate the need for macros and replace conditional processing with inline statements.

Example: %if statement used to conditionally execute blocks of code.

```
%macro DoThis;  
%if <condition> %then %do;  
    * statements for true;  
%end;  
%else %do;  
    * statements for false;  
%end;  
run;%mend;  
%DoThis
```

can be replaced with this:

```
%sysfunc(ifc(<condition>  
    , * statements for true;  
    , * statements for false;  
))
```

---

---

## Functions

### Overview

This trick combines the data step function ifc and the macro function sysfunc.

- data step function ifc
  - macro function sysfunc
  - combining sysfunc and ifc
-

---

## Data Step Function IFC

---

### Overview

The ifc function has four parameters:

1. a logical expression
2. character value returned when true
3. value returned when false
4. value returned when missing, which is optional

### Example

This example shows the function in a data step.

```
1 DATA Work.TestIfc;
2 attrib N length = 4
3 Value length = $ %length(missing);
4 do N = ., 0, 1, -1;
5     Value = ifc(N
6         , 'true'
7         , 'false'
8         , 'missing'
9     );
10     output;
11 end;
12 stop;
13 run;
14
15 Proc Print data = Work.TestIfc;
```

### Output

1	Obs	N	Value
2	---	--	-----
3	1	.	missing
4	2	0	false
5	3	1	true
6	4	-1	true

Note that minus one is true (or not(false)).

---

---

## Macro Function SysFunc

---

### Overview

The macro function `sysfunc` provides a method to access certain data step functions and the values of options.

### Example

This example shows the combination of `sysfunc` and `getoption` writing a note to the log with the value of the option `source2`.

```
1 options nosource2;
2 %Put option source2 is %sysfunc(getoption(source2));
3 option source2 is NOSOURCE2
4 options source2;
5 %Put option source2 is %sysfunc(getoption(source2));
6 option source2 is SOURCE2
```

---

---

## Combining SysFunc and IFC

### Macro Statement PUT with IFC

This example contains the previous usage of `sysfunc` and `getoption`. Its purpose is to show that the evaluation of the logical expression can produce conditional text.

```
1 1 options nosource2;
2 2 %Put option source2 is
3 3 %sysfunc(ifc(%sysfunc(getoption(source2)) eq SOURCE2
4 4 ,True
5 5 ,False
6 6 ,Missing
7 7 ) );
8 option source2 is False
```

Note that the `sysfunc(ifc(...))` is within the `%put` statement whose ending semicolon is on log line 7.

### IFC producing PUT

The promise of `sysfunc(ifc(...))` is to generate statements, based on the evaluation. This example shows the first test: a complete `%put` statement as the result of the evaluation.

### The Error

```
1 1 options nosource2;
2 2 %sysfunc(ifc(%sysfunc(getoption(source2)) eq SOURCE2
3 3 ,%Put True;
4 True
5 4 ,%Put False;
6 False
7 5 ,%Put Missing;
8 Missing
9 6 ))
```

What happened here? All the macro `%put` statements were executed!

### Nrstr of PUT

This problem is the same as with `call execute`, which is explained by Fehd and Carpenter [4, sgf2007.113]. The statements must be wrapped in macro function `nrstr` (No Rescan String) which delays their execution until after the `ifc` condition has been evaluated.

```
1 1 options nosource2;
2 2 %sysfunc(ifc(%sysfunc(getoption(source2)) eq SOURCE2
3 3 ,%nrstr(%Put True;)
4 4 ,%nrstr(%Put False;)
5 5 ,%nrstr(%Put Missing;)
6 6 ) )
7 False
```

Note that the `sysfunc(ifc(...))` is not a statement; the function calls are completed by the two closing parentheses on log line 6.

---

---

## Usage Examples

---

### Overview

These are common tests:

- boolean: zero, one or greater than one
- comparison of two values
- existence

### Boolean

Many functions return either the set of boolean values — (0,1) — or zero and a negative or positive integer which evaluates as true. Remember that true is not false, see p. 3.

```
0,1 : %sysfunc(cexist(&Catalog.))
```

```
0, ge 1 : &Nobs.
```

```
-1,0, ge 1 : %sysfunc(attrn(&dsid., nobs))
```

### Comparison

Any logical expression can be used for the first argument of the `ifc` function.

- `&NobsData1. eq &NobsData2.`
- `&Nobs. gt 100`
- `%sysfunc(fileref(&FileRef.)) eq 0`
- `%upcase(%substr(&Type.,1,1)) eq C or &Type. eq 2`

There are pitfalls when comparing user-supplied values; refer to Fehd [2, sas-wiki.Cond-Exec-Global-Stmnts] for examples and fixes.

### Test Suite

Each of the following programs has been tested as a parameterized include file, illustrated by this example:

```
1 options source2;
2 Proc Format library = Work; value Test 0='zero';run;
3 %Let Catalog = Work.Formats;
4 %Include Project(assert-cexist-catalog);
5 %Let Catalog = Library.Formats;
6 %Include Project(assert-cexist-catalog);
```

Both the program `*.sas` and its test `*-Test.sas` are available in a .zip at Fehd [2, sas-wiki.Cond-Exec-Global-Stmnts]

---

---

## Calling Routines

### Branching with Includes

Evaluation of the logical expression can be used to branch to the execution of either of two programs.

```
1  * cond-inc-which;
2  options source2;
3  %let Data = sashelp.Class;
4  %*let Data = sashelp.ClassX;* test not exist;
5  %let dsid = %sysfunc(open(&Data. ));
6  %let Nobs =
7  %sysfunc(ifc(&DsId.
8              ,%nrstr(
9                  %sysfunc(attrn(&dsid.,nobs));
10                 %let rc  = %sysfunc(close(&dsid.)); )
11              ,%nrstr(0) ));
12 %sysfunc(ifc(&Nobs.
13             ,%nrstr(%Include Project(cond-inc-1);)
14             ,%nrstr(%Include Project(cond-inc-0);) ))
15
1  %Put Note2: file is cond-inc-1;
2
1  %Put Note2: file is cond-inc-0;
```

### Procedure By Type

Evaluation of the logical expression can be used to branch to the execution of either of two sets of statements.

The elaborate test of the value of Type ensures that this code can be called with a list processing data set created by either Proc Contents whose Type are in (1,2) and Proc SQL whose Type are in ('char','num').

```
1  title3 "proc-type of &Libname..&Memname.;";
2  %sysfunc(ifc( %upcase(%substr(&Type.,1,1)) eq C
3              or %Type eq 2
4              ,%nrstr(Proc Freq data = &Libname..&Memname.;
5                      tables &Name.;)
6              ,%nrstr(Proc Means data = &Libname..&Memname.;
7                      var &Name.;) ))
8  run;
```

---

Continued on next page.

## Calling Proc-Type

This code is similar to the conditional processing in the SmryEachVar data review suite Fehd [3, sgf2008.003].

```
1  options source2;
2  %Let Libname = sashelp;
3  %Let Memname = Class;
4  %Let Name     = Sex;
5  %Let Type     = C;
6  %Include Project(proc-type);
7
8  %Let Name     = Height;
9  %Let Type     = 1;
10 %Include Project(proc-type);

1  options source2;
2  %Let Libname = sashelp;
3  %Let Memname = Class;
4  Proc SQL; select catt('%let Name =',Name,',';
5                  ,      '%let Type =',Type,',';
6                  ,'%Include Project(proc-type)')
7                  into :List separated by ';';
8                  from Dictionary.Columns
9                  where Libname eq "%upcase(&Libname.)"
10                 and Memname eq "%upcase(&Memname.)"
11                 and Memtype eq 'DATA';
12                 quit;
13 &List.;

1  options source2;
2  %Let Libname = sashelp;
3  %Let Memname = Class;
4  Proc Contents data = &Libname..&Memname.
5                  noprint
6                  out  = Work.Contents;
7  DATA _Null_;
8  attrib Stmnt length = $72;
9  do until(EndoFile);
10     set Work.Contents end = EndoFile;
11     Stmnt = catt('%let Name =',Name,',';');
12     link ExecStmt;
13     Stmnt = catt('%let Type =',Type,',';');
14     link ExecStmt;
15     Stmnt = '%Include Project(proc-type)';
16     link ExecStmt;
17     end;
18 return;
19 ExecStmt: call execute(catt('%nrstr(',Stmnt,')'));
20 return;
21 stop;
22 run;
```

Continued on next page.

## Number of Rows

The `attrn` function can be used to get the number of observations of a data set. The data set must be opened before the function can return the value. If the data set exists then the function can be applied; otherwise when the open of the data set fails, it is not necessary to close the data set.

The values returned by `attrn(..., nobs)` are in:

-1 : for a view

0 : empty

ge 1 : number of rows

```
1 %let dsid = %sysfunc(open(&Data. ));
2 %let Nobs =
3 %sysfunc(ifc(&Dsid.
4             ,%nrstr(
5                 %sysfunc(attrn(&dsid.,nobs));
6                 %let rc = %sysfunc(close(&dsid.)); )
7             ,%nrstr(0) ));
8 %sysfunc(ifc(&Nobs.
9             ,%nrstr(%Put Note2: Data &Data. Nobs: &Nobs.);
10            ,%nrstr(%Put %sysfunc(sysmsg());
11                    endSAS;) ))
```

As shown above on p. 3 minus one evaluates as true.

```
21 DATA Work.ClassView / view = Work.ClassView;
22 set sashelp.class(obs=1);
23 run;
```

NOTE: DATA STEP view saved on file WORK.CLASSVIEW.

```
24 %Let Data = Work.ClassView;
25 %Include Project(assert-nobs);
Note2: Data Work.ClassView available Nobs: -1
```



---

## Assertions

### Overview

Assertions are tests at the beginning of a program that determine whether the program input parameters are valid. The assertion of existence follows the rules of simple boolean evaluation: the functions return zero or one.

### Example of Usage

```
* name: MyProgram;  
* confirm input exists;  
%let Object = ObjectName;  
%Include Asserts(ExistObject);* false==endSAS;  
*processing;
```

### Exist Catalog

The `cexist` function can be used to test the existence of a catalog.

```
1 %sysfunc(ifc(%sysfunc(cexist(&Catalog.))  
2 ,%nrstr(%Put Note2: Catalog &Catalog. exists;)  
3 ,%nrstr(%Put Note3: not exist &Catalog.;  
4 endSAS;) ))
```

### Exist Data

The `exist` function can be used to test whether a data set exists.

```
1 %sysfunc(ifc(%sysfunc(exist(&Data.))  
2 ,%nrstr(%Put Note2: Data &Data. exists;)  
3 ,%nrstr(%Put Note3: not exist &Data.;  
4 endSAS;) ))
```

### Exist Fileref

The `fexist` function can be used to test the existence of a fileref which is a directory-specification or folder.

```
1 %sysfunc(ifc(%sysfunc(fexist(&FileRef.))  
2 ,%nrstr(%Put Note2: FileRef &FileRef. exists;)  
3 ,%nrstr(%Put Note3: not exist &FileRef.;  
4 endSAS;) ))
```

### Exist File

The `fileexist` function can be used to test the existence of a file-specification.

```
1 %sysfunc(ifc(%sysfunc(fileexist(&FileNameExt.))  
2 ,%nrstr(%Put Note2: FileNameExt &FileNameExt. exists;)  
3 ,%nrstr(%Put Note3: not exist &FileNameExt.;  
4 endSAS;) ))
```

---

Continued on next page.

---

## Exist Fileref and File

The `fileref` function is a special case. Instead of returning one as success and zero as failure, it returns this set of values:

-1 : `fileref` exists but file associated with the `fileref` does not exist

0 : `fileref` and external file both exist

+1 : `fileref` not assigned

Thus success is defined as `eq 0!`

```
1 %sysfunc(ifc(%sysfunc(fileref(&FileRef.)) eq 0
2           ,%nrstr(%Put Note2: FileRef &FileRef. exists;))
3           ,%nrstr(%Put Note3: not exist &FileRef.;
4           endSAS;))
```

For an example of usage of function `fileref` see the OnLine Doc for function `finfo`.

## Summary of Assertion Usage

Each of these examples show the failure as sufficient to terminate the program — `endSAS;` In practice consider adding other programs which provide information written to a log or data set. Nelson et al. [6, sesug2003.001] discuss a system of event management where subroutines log program completion and e-mail notifications of failure are sent to interested users.

---

---

## Conclusion

The combination of functions `sysfunc` and `ifc` can be use to eliminate the need for macros where only conditional processing is needed.

---

## Further Reading

<b>Download</b>	The code examples in this paper are available on the SAS community wiki at Fehd [2, sas-wiki.Cond-Exec-Global-Stmnts]
<b>Assertions</b>	Nelson et al. [6, sesug2003.001] describe assertions in an event management and notification system.
<b>Call Execute and Nrstr</b>	Fehd and Carpenter [4, sgf2007.113] demonstrate the problem of using call execute to generate macro calls and how using function <code>nrstr</code> solves that problem.
<b>Conditional Execution of Includes</b>	Fehd [1, sugi25.038] provides a trick for conditional execution of included programs.
<b>Macro IFF</b>	Henderson [5, saspress.60282] provides a macro to get around the problem of user-supplied values in comparisons.
<b>Using Parameterized Includes</b>	Fehd [3, sgf2008.003] provides a data review suite of 80 programs with modules, routines and subroutines of parameterized includes.

## References

- [1] Ronald Fehd. The Writing for Reading SAS Style Sheet: Tricks, traps, tips, and templates, from SAS-L's Macro Maven. In *Proceedings of the 25th SAS User Group International Conference*, 2000. URL <http://preview.tinyurl.com/4rug92>. Coders Corner, 4 pp.; topics: avoiding common mistakes when writing macro statements, recommendations; info: conditional execution of includes.
- [2] Ronald J. Fehd. Conditionally executing global statements. In *SAS Community Wikipedia*, 2008. URL [http://www.sascommunity.org/wiki/Conditionally\\_Executing\\_Global\\_Statements](http://www.sascommunity.org/wiki/Conditionally_Executing_Global_Statements). topics: combining functions sysfunc and ifc; info: example assertions, caveats on logical comparisons.
- [3] Ronald J. Fehd. SmryEachVar: A data-review routine for all data sets in a libref. In *Proceedings of the SAS Global Forum Annual Conference*, 2008. URL <http://preview.tinyurl.com/5nw63d>. Applications Development, 24 pp.; keywords: call execute, data review, dynamic programming, includes, list processing, nrstr, routines, source2, subroutines, sysparm; topics: data review using proc freq; info: utilities to repair missing elements (formats, labels) in data structure.
- [4] Ronald J. Fehd and Art Carpenter. List processing basics: Creating and using lists of macro variables. In *Proceedings of the SAS Global Forum*, 2007. URL <http://preview.tinyurl.com/6mmqpj>. Hands On Workshop, 20 pp.; comparison of methods: making and iterating macro arrays, scanning macro variable, writing calls to macro variable, write to file then include, call execute and nrstr; 11 examples, bibliography.
- [5] Don Henderson. *Building Web Applications with SAS/IntrNet: A Guide to the Application Dispatcher*. SAS Institute, 2006. URL <http://www.sas.com/apps/pubscat/bookdetails.jsp?catid=1&pc=60282>. ISBN 978-1-59994-189-9.
- [6] Greg Barnes Nelson, Danny Grasse, and Deborah Pine. Automated testing and real-time event management. In *Proceedings of the SouthEast SAS User Group Conference*, 2003. URL [http://www.thotwave.com/Document/papers/SESUG\\_03/EventSys/eventmngmt.pdf](http://www.thotwave.com/Document/papers/SESUG_03/EventSys/eventmngmt.pdf). 6 pp. topics: testing, definition of events, batch scheduler, program agent, execution monitor, real-time notification.

## Author Information

**Author: Ronald J. Fehd**  
**Centers for Disease Control**  
**4770 Buford Hwy NE**  
**Atlanta GA 30341-3724**

<mailto:RJF2@cdc.gov>

about the author:

education:	B.S. Computer Science, U/Hawaii,	1986
	SAS User Group conference attendee since	1989
	SAS-L reader	since 1994
experience:	programmer: 20+ years	
	data manager at CDC, using SAS: 17+ years	
	author: 30+ SUG papers	
SAS-L:	author: 5,000+ messages to SAS-L since	1997
	Most Valuable SAS-L contributor:	2001, 2003

## SAS Institute Information

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. In the USA and other countries ® indicates USA registration.